Introduction to Lab 5

In Calculus III, we saw that space curves are typically defined by a set of parametric equations in time. For example, a helix in three dimensions may be defined by the map F below, or the infamous "pringle" defined by the map G:

$$F(t) = \begin{bmatrix} \cos(t) \\ \sin(t) \\ t \end{bmatrix} \qquad G(t) = \begin{bmatrix} \cos(t) \\ \sin(t) \\ cos(2t) \end{bmatrix}$$

In any event, the main point is that when the domain is a single variable (time t), the result in two or three dimensions is a *curve*. In the case of a curve in two dimensions, $y = f(x)$, it is easy to parametrize it by taking $x = t$:

$$y = f(x) \quad \Leftrightarrow \quad \begin{bmatrix} t \\ f(t) \end{bmatrix}$$

Alternatively, a **surface** can be defined directly by

$$z = f(x, y)$$

or x, y and z can be functions of *two* variables, s, t. An easy way to parametrize a surface is to let $x = s$ $y = t$, then $z = f(s, t)$. For example:

$$z = x^2 + y^2 \quad \Leftrightarrow \quad \begin{bmatrix} s \\ t \\ s^2 + t^2 \end{bmatrix}$$

We can get more exotic- For example, how might we parametrize a sphere of radius 1? The surface of the sphere can be represented by two angles- ϕ and θ, where ϕ measures the angle downward from the north pole, and θ measures the angle you rotate parallel to the equator. For a given fixed ϕ, the set of points drawn out by taking all θ would be a circle parallel to the xy plane- If we knew the radius R, then

$$x = R\cos(\theta) \qquad y = R\sin(\theta)$$

In the xz plane, ϕ is measured from the north pole. Along a circle in this plane, (R, z) would be given by:

$$R = \cos(\pi/2 - \phi) = \sin(\phi) \qquad z = \sin(\pi/2 - \phi) = \cos(\phi)$$

Put it all together, and x, y and z are parametrized as:

$$x = \sin(\phi)\cos(\theta) \qquad y = \sin(\phi)\sin(\theta) \qquad z = \cos(\phi)$$

Maple Exercises

Try these Maple exercises out to get accustomed to parametric plots in two and three dimensions.

1. Plot a two dimensional parametric plot for $F(t) = < \cos(t)/t, \sin(t)/t >$ for t from 1/2 to 20.

   ```
   with(plots):
   plot([cos(t)/t,sin(t)/t,t=0.5..20]);
   ```

2. Plot the trefoil knot, given parametrically by:

   $$x(t) = -10\cos(t) - 2\cos(5t) + 15\sin(2t), \quad y(t) = -15\cos(2t) + 10\sin(t) - 2\sin(5t)$$

   $$z(t) = 10\cos(3t)$$

   Rather than plotting this as a parametric curve, we can thicken it into a tube using `tubeplot`:

   ```
   xt:=-10*cos(t)-2*cos(5*t)+15*sin(2*t);
   yt:=-15*cos(2*t)+10*sin(t)-2*sin(5*t);
   zt:=10*cos(3*t);
   tubeplot([xt,yt,zt],t=0..2*Pi,radius=1);
   ```

3. Plot the pringle (parametric in three dimensions needs the `spacecurve`). If you've already entered `with(plots):`, you don't need to do it again.

   ```
   with(plots):
   spacecurve([cos(t),sin(t),cos(2*t)],t=0..2*Pi,axes=boxed);
   ```

4. Plot the upper half sphere (unit radius):

   (a) Method 1: Using $z = f(x, y)$, we would have

   $$z = \sqrt{1 - x^2 - y^2}$$

   but this method leaves some gaping holes about the equator:

   ```
   plot3d(sqrt(1-x^2-y^2),x=-2..2,y=-2..2);
   ```

   This could be worked over by setting the grid higher:

   ```
   plot3d(sqrt(1-x^2-y^2),x=-2..2,y=-2..2,grid=[80,80]);
   ```

(b) Method 2: Using angles so that x, y and z are all functions of two other variables- These are simply the spherical coordinate equations:

$$x = \sin(\phi)\cos(\theta) \quad y = \sin(\phi)\sin(\theta) \qquad z = \cos(\phi)$$

We think of ϕ as the angle coming down from the North Pole, and θ as the angle in the $x - y$ plane. To get a half circle, we plot the surface taking $0 \leq \phi \leq \pi/2$ and θ runs all the way from 0 to 2π:

```
xt:=sin(phi)*cos(theta);
yt:=sin(phi)*sin(theta);
zt:=cos(phi);
plot3d([xt,yt,zt],phi=0..Pi/2,theta=0..2*Pi,scaling=constrained);
```